
log using "e:\path analyisis.log", replace

* sembuilder

*******************************
* example 42g : One- and two-level mediation models (multilevel)
*******************************
use http://www.stata-press.com/data/r15/gsem_multmed, clear
gen hi_perm = 1
replace hi_perm = 0 if perform < 5.005317 
label variable hi_perm "high performer"
label define hi_perm 1 "high performer" 0 "low performer"
label value hi_perm hi_perm

********************
* Single-level mediation 
********************

*multiple regression with the reg command
reg perform satis suppor

* Muleiple Regression wtih the sem command
sem (perform <- satis support)
estat gof, stats(all)
estat teffects

* Muleiple Regression wtih the sem command and no covariance beween Satis and support
sem (perform <- satis support), cov(satis*support@0)
estat gof, stats(all)
estat teffects

* Muleiple Regression wtih the sem command and no covariance beween Satis and support, and the equal 
* regression coefficient for satis and support

sem (perform <- satis@b support@b), cov(satis*support@0)
estat gof, stats(all)
estat teffects

* Muleiple Regression wtih the sem command and no covariance beween Satis and support, and fix the regression 
* coefficient of satis to perform as 0.78

sem (perform <- satis@0.78 support), cov(satis*support@0)
estat gof, stats(all)
estat teffects

* Muleiple Regression wtih the sem command and covariance beween Satis and support

sem (perform <- satis support) (satis <- support)
estat gof, stats(all)
estat teffects

******************
* Multiple-group comparison
******************

sem (perform <- satis support) (satis <- support), group(hi_perm) 
sem (perform <- satis support) (satis <- support), group(hi_perm) ginvariant(all)
estat teffects

***********************************
* one-level model with gsem
***********************************

gsem (perform <- satis support) (satis <- support)
gsem, coeflegend
nlcom _b[perform:support]+_b[perform:satis]*_b[satis:support]

***********************
* Two-level model with gsem
************************

gsem (perform <- satis support M1[branch]) (satis <- support M2[branch]), cov(M1[branch]*M2[branch]@0)

nlcom _b[perform:satis]*_b[satis:support]
nlcom _b[perform:support]+_b[perform:satis]*_b[satis:support]


